

宇宙航空研究開発機構研究開発報告

JAXA Research and Development Report

回転ブレードの自由渦後流・翼素理論による全状態空力特性の概算

Rough Estimation for All-State Aerodynamic Characteristics on Rotating Blades
by Free Vortex Wake and Blade Element Theory

奥山 政広, 小林 宙

OKUYAMA Masahiro, KOBAYASHI Hiroshi

2022年8月

宇宙航空研究開発機構

Japan Aerospace Exploration Agency

目次

概要	1
1. はじめに	2
2. 数値計算モデル	3
2.1 誘導速度	4
2.2 自由渦後流・翼素理論による定式化	5
3. 対象ブレードと数値計算結果	12
3.1 R.A.F.6翼型を持つ3 feet プロペラ	14
4. まとめ	19
参考文献	19
付録	20

回転ブレードの自由渦後流・翼素理論による全状態空力特性の概算

奥山 政広^{*1}, 小林 宙^{*1}

Rough Estimation for All-State Aerodynamic Characteristics on Rotating Blades by Free Vortex Wake and Blade Element Theory

OKUYAMA Masahiro^{*1}, KOBAYASHI Hiroshi^{*1}

ABSTRACT

Focusing on the aerodynamic characteristics of the windmill states with positive and negative thrust, a numerical calculation program was developed using free vortex wake and airfoil element theories in order to easily estimate the all-state aerodynamic characteristics of rotating blades. The rotating blade is assumed to be a propeller blade with a variety of inflow conditions. The all-states of the rotating blades are the windmill state with negative thrust, the normal state, the static thrust state, and the windmill state with positive thrust wind for relatively large values of the advance ratio through zero to negative values.

The numerical calculation program provides a rough estimate of the aerodynamic characteristics for the parameter variable pitch angle and the advance ratio, which can be varied in the positive and negative ranges.

The calculation results by this program are compared with the experimental results under normal state. In addition, the results are extended to windmill states with positive and negative thrust, which are not found in the experimental results, to give a rough estimate of the aerodynamic characteristics.

Keywords: Propeller, Rotor Blade, Free Vortex Wake, Blade Element Theory, Aerodynamic Characteristics, Negative Power

概 要

正・負推力の風車状態の空力特性に着目して、回転ブレードの全状態空力特性を簡便に概算するため、自由渦後流・翼素理論により数値計算プログラムを作成した。回転ブレードは、多彩な流入条件を伴うプロペラのブレードを想定している。回転ブレードの全状態は、進行率の比較的大きな値から 0 を経て負の値までに対する、負推力の風車状態、通常状態、静止推力状態、正推力の風車状態である。

数値計算プログラムは、パラメータである可変ピッチ角、および正負範囲で変化させる進行率に対して空力特性が概算できる。

このプログラムによる計算結果は、通常状態において、実験結果と比較している。さらに、実験結果にない正・負推力の風車状態へ拡張して空力特性の概算を示す。

^{*} 2022 年 6 月 1 日受付 (Received June 1, 2022)

^{*1} 航空技術部門 (Aviation Technology Directorate)

主な記号

B ブレード枚数	V_t ブレード回転面の接線方向対気速度
c ブレードの翼弦長	V_x x 軸対気速度
c_l 翼型の揚力係数	V_z z 軸対気速度
c_d 翼型の抗力係数	w_a 回転軸方向誘導速度
C_T 推力係数 = $T/(\rho n^2 D^4)$	w_t 回転面接線方向誘導速度
C_P 吸収パワー係数 = $P/(\rho n^3 D^5)$	α, ALP 迎え角 = $\beta - \phi$
D 回転ブレードの直径	α_e 有効迎え角 = $\beta - (\phi + \alpha_i)$
FM フィギュア・オブ・メリット	β ピッチ角 = $\beta_t + \beta_v$
I 1回転の分割数	β_t ねじり分布ピッチ角
J ブレード束縛渦の分割接点数	β_v, BETv 可変ピッチ角
J_V 進行率 = $V/(nD)$	Γ, GAM 循環
M 回転回数	Δt 時間ステップ
n, N 回転数[Hz], [r.p.m.]	$\Delta \Gamma$ 自由渦の強さ(循環)
p/D ピッチ直径比	η, ETA B 枚ブレードの推進効率 = VT/P
Q B 枚ブレードの吸収トルク	$1/\eta, 1/\text{ETA}$ B 枚ブレードの逆効率
R ブレード半径 ($D/2$)	θ, THE チルト角
r ブレード半径方向で回転中心からの位置	ρ 飛行高度の大気密度
r_c 渦核の半径	$\phi + \alpha_i$ 流入角・誘導迎え角
T B 枚ブレードの推力	ψ 回転角
P B 枚ブレードの吸収パワー	Ω 回転角速度
V ブレードの回転軸方向対気速度	

1. はじめに

大枠として、回転ブレードにおいて、諸条件を与えてブレード形状を設計する方法が逆問題ならば、ここで用いる運動量・翼素理論による数値計算の方法は、ブレード形状とその翼型空力特性を与えてブレードの空力特性を得る順問題に位置付けられる。本報告の主目的は、既存や想定で与えたブレード形状に対して、負の吸収パワー状態にある広範囲な進行率の空力特性が概算できる数値計算の方法を示ことである。負の吸収パワー状態は、進行率の相対的に大きい負推力の風車状態、および負の可変ピッチ角で負の進行率をとる正推力の風車状態がある。

プロペラの空力特性値の計算における誘導速度の数値計算方法は、運動量理論と翼素理論を組み合わせた「運動量・翼素理論」¹⁾や、プラントルの翼端損失係数 (Prandtl's tip loss factor) という関数^{2),3)}を使用した翼素理論である「渦理論」¹⁾などがある。また、ダクテッドプロペラの最適設計における方法は、揚力線の分割点から螺旋状に放出渦を直線近似要素で渦面移動速度を伴い放出させ、ビオ・サバールの法則 (Biot-Savart law) により揚力線のコントロールポイントで誘導速度が計算されている⁴⁾。

この自由渦後流・翼素理論は、風力タービン (wind turbine) で用いられる揚力線自由渦後流法 (lifting-line free vortex wake method^{5),6)} と同じ様な方法であり、より単純である。この自由渦後流・翼素理論は、与えたブレード形状の揚力線として分割した束縛渦の接点から、時間ステップ毎に排出される直線近似要素から成る自由渦後流系が、束縛渦の代表点へビオ・サバールの法則により計算される誘導速度を生じさせるだけでなく、ビオ・サバールの法則により計算される自らの誘導速度場および対気速度から、自由渦後流系の要素位置を時間ステップ毎に変位させ、新しい自由渦後流系が成長していく。時間ステップの繰り返しを何周か重ねると、束縛渦の強さが一定な値に落ち着く。

回転ブレードの翼型は、負推力の風車状態になる大きい進行率から、静止推力状態の 0 進行率を経て、負側に下げた可変ピッチ角で正推力の風車状態になる負の進行率 (逆向き対気速度) まで、進行率が流

入角を決めるため、幅広い迎え角をとる。それゆえ、翼型の空力特性を大きい迎え角まで与えることにより、広範囲な進行率に対応が可能となる。回転ブレードの空力特性において、誘導速度場で変位する自由渦後流は実在現象に近い物理モデルであり、この自由渦後流・翼素理論はその数学モデルになる。この自由渦後流・翼素理論の計算プログラムは、単純なアルゴリズムであり、繰り返し処理のネストでいくらか計算時間が掛かるものの、与えた回転ブレードで 1 点の進行率に対する推力係数およびパワー係数の空力特性値を、簡便に概算できる。

2. 数値計算モデル

数値計算のモデルは、自由渦後流を用いた翼素理論である。翼素理論は、回転する翼であるブレードを翼幅方向に離散化し、分割された翼素を設ける。翼素は 2 次元翼である翼型であり、揚力線として束縛渦を一般的な空力中心の 1/4 翼弦位置に置き、束縛渦の翼素両端を接点、および、接点間の中央を代表点にとる。ただし、各翼素の代表点の位置が前後に食い違っても、近似してブレード翼素の翼型重心を貫くピッチ角変更軸へ一致させる。したがって、ブレードのモデルは、ピッチ角変更軸へ直線に配置された揚力線の束縛渦であり、接点で分割された接点間の midpoint に離散束縛渦の代表点を置く。

ブレードは 3 次元翼として、回転する離散束縛渦のそれぞれ接点から自由渦を流出し、時間ステップ毎に接点を伴って排出される直線近似要素が連なり自由渦後流を形成する。それら自由渦後流の系による誘導速度が、それぞれ代表点でビオ・サバールの法則から計算される。それぞれの代表点で、誘導速度ベクトルと流入速度ベクトルを合成すると、翼素のピッチ角から翼型に対する有効迎角が求まり、与えている風洞試験データの空力特性から翼型の有効迎え角に対する揚力および抗力係数が得られる。この揚力係数が、クッタ・ジューコフスキーの定理 (Kutta-Joukowski theorem) により、翼素の循環へ変換される。

束縛渦の各接点から時間ステップ毎に流出する直線近似要素が連なる自由渦の強さは、時間ステップ毎、その接点が接する束縛渦の強さの差分になる。両端に接点を持つ自由渦の直線近似要素は、ブレードが回転してある時間ステップ後に、前の位置の接点において、自由渦後流系によるビオ・サバールの法則を用いた誘導速度ベクトルが決まる。この誘導速度ベクトルと対気速度ベクトルの合成速度ベクトルへ時間ステップを掛ければ、位置ベクトルが求まり、時間ステップ後の位置ベクトルとして、前の位置と直線で結び自由渦後流となる。全接点において、この位置ベクトルを計算し、また、形成された自由渦後流系による束縛渦を再計算して、次の時間ステップに進み、接点と自由渦の線分を直線近似要素として付け加え、新たな自由渦後流系を形成していく。

空力特性値は、ある程度の回転後にブレードの空力特性の値が安定するので、それを計算結果とする。ブレードの空力特性である推進方向の推力係数および回転方向のパワー係数は、翼素に作用する揚力係数および抗力係数の力ベクトル合成から求める。ある程度の回転後に回転角が 0° に戻った位置での特性値を計算結果とするため、ピッチ角を取った場合でも回転角の位置による影響を受けず、この値を代表値にとる。

誘導速度の計算は、ビオ・サバールの法則の式へ渦核を持つ渦モデルを組み入れる。また、計算において、回転ブレードの回転方向は、全てが、回転軸方向の前方へ向き時計回りの設定になる。それゆえ、ブレードにねじり分布ピッチ角 (回転面基準) が設けられているなら、通常の場合で、ねじり分布ピッチ角の大きいブレード根元から先端に向かい反時計方向へねじり下げられている。

2.1. 誘導速度

原点 O の直角座標 x, y, z に基づいた図 2.1.1 のベクトル図において、位置ベクトル \mathbf{r}_A の点 A から \mathbf{r}_B の点 B までの線分 l_{AB} で循環 Γ を持つ渦糸により \mathbf{r}_P の点 P に誘導される速度 \mathbf{w} は、微小部分にビオ・サバールの法則を適用して、微小渦糸の要素 $d\mathbf{l}$ から P 点へのベクトルを \mathbf{r} すれば誘導速度 $d\mathbf{w}$ が

$$d\mathbf{w} = \frac{\Gamma}{4\pi} \frac{d\mathbf{l} \times \mathbf{r}}{|\mathbf{r}|^3} \quad (2.1.1)$$

と表される。ビオ・サバールの法則は、電流により生ずる電場の強さを示す電磁気現象から流体现象へのアナロジーとして使用されている。ただし、流体力学の基礎的な方程式からビオ・サバールの法則を導くことができる⁷⁾。

図 2.1.1 のベクトル図で、点 A から点 P 、および点 B から点 P のベクトル \mathbf{r} をそれぞれ \mathbf{r}_{AP} および \mathbf{r}_{BP} とすれば、 l_{AB} 、 \mathbf{r}_{AP} 、および \mathbf{r}_{BP} は、三角形の平面を形作る。また、 l_{AB} と \mathbf{r} の成す角度を θ にとり、 A 点で θ_A および B 点で θ_B とする。さらに、点 P から l_{AB} へ下す垂線の長さを h とする。それゆえ、 r が式(2.1.3)の関係式で示される。




図 2.1.1 渦糸線分 AB により点 P で誘導される速度の関係ベクトル図

$$r \sin(\pi - \theta) = r \sin \theta = h \quad (2.1.2)$$

$$r = \frac{h}{\sin \theta} \quad (2.1.3)$$

引き続いて、 l_{AB} の該当長さ l から、 h との関係式が

$$l \tan(\pi - \theta) = -l \tan \theta = h \quad (2.1.4)$$

$$l = -\frac{h}{\tan \theta} \quad (2.1.5)$$

となる。この式(2.1.5)から θ に対する微分が

$$dl = \frac{h}{\sin^2 \theta} d\theta \quad (2.1.6)$$

になる。それゆえ、微小部分による誘導速度の大きさ $d\mathbf{w}$ は、式(2.1.3)の r および式(2.1.6)の dl を式(2.1.1)へ代入して、式(2.1.7)に帰結する。

$$\begin{aligned} d\mathbf{w} &= \frac{\Gamma}{4\pi} \frac{d\mathbf{l} h}{r^3} \\ &= \frac{\Gamma}{4\pi} \frac{\sin^3 \theta}{h^3} h \frac{h}{\sin^2 \theta} d\theta \\ &= \frac{\Gamma}{4\pi h} \sin \theta d\theta \end{aligned} \quad (2.1.7)$$

この式を、 θ に対して θ_A から θ_B まで積分すれば、誘導速度の大きさ \mathbf{w} が

$$w = \frac{\Gamma}{4\pi h} (\cos \theta_A - \cos \theta_B) \quad (2.1.8)$$

と求まる。

以上の式は、ベクトル表記すれば式(2.1.9)で示される。

$$\begin{aligned} \begin{bmatrix} w_x \\ w_y \\ w_z \end{bmatrix} &= \frac{\Gamma}{4\pi h} (\cos \theta_A - \cos \theta_B) \begin{bmatrix} e_x \\ e_y \\ e_z \end{bmatrix} \\ \begin{bmatrix} e_x \\ e_y \\ e_z \end{bmatrix} &= \frac{\mathbf{l}_{AB} \times \mathbf{r}_{AP}}{|\mathbf{l}_{AB} \times \mathbf{r}_{AP}|} = \frac{\mathbf{r}_{AP} \times \mathbf{r}_{BP}}{|\mathbf{r}_{AP} \times \mathbf{r}_{BP}|} \\ \mathbf{l}_{AB} &= \mathbf{r}_B - \mathbf{r}_A \\ \mathbf{r}_{AP} &= \mathbf{r}_P - \mathbf{r}_A \\ \mathbf{r}_{BP} &= \mathbf{r}_P - \mathbf{r}_B \\ \cos \theta_A &= \frac{\mathbf{l}_{AB} \cdot \mathbf{r}_{AP}}{|\mathbf{l}_{AB}| |\mathbf{r}_{AP}|} \\ \cos \theta_B &= \frac{\mathbf{l}_{AB} \cdot \mathbf{r}_{BP}}{|\mathbf{l}_{AB}| |\mathbf{r}_{BP}|} \\ h &= \frac{|\mathbf{l}_{AP} \times \mathbf{r}_{AP}|}{|\mathbf{l}_{AB}|} = \frac{|\mathbf{r}_{AP} \times \mathbf{r}_{BP}|}{|\mathbf{l}_{AB}|} \end{aligned} \quad (2.1.9)$$

ベクトルの関係式をまとめて整理すると、ビオ・サバールの法則から、循環 Γ を持つ渦糸線分 AB による点 P での誘導速度 \mathbf{w} は

$$\mathbf{w} = \frac{\Gamma}{4\pi} \frac{\mathbf{r}_{AP} \times \mathbf{r}_{BP}}{|\mathbf{r}_{AP} \times \mathbf{r}_{BP}|^2} \left[\mathbf{l}_{AB} \cdot \left(\frac{\mathbf{r}_{AP}}{|\mathbf{r}_{AP}|} - \frac{\mathbf{r}_{BP}}{|\mathbf{r}_{BP}|} \right) \right] \quad (2.1.10)$$

とベクトル式で表示できる。

誘導速度の大きさは渦糸中心からの距離の逆数 $1/h$ に比例するので、 $h \rightarrow 0$ における無限大を避けるため、文献で取り扱われている方法⁸⁾を用いて、式(2.1.11)のように渦糸を半径 r_c の核を持つ渦と置き換え、設定値 d を 1 あるいは 2 にとり渦の速度分布⁹⁾を定めれば、誘導速度 \mathbf{w} の式が書き改められる。ここでは、ラム・オーセンの渦モデルを近似する $d = 2$ に選ぶ¹⁰⁾。渦核の半径 r_c の値は、経過時間に関係なく一定値として、ブレード半径方向の特性値分布で計算結果が暴れない程度に設定する。

$$\mathbf{w} = \frac{\Gamma}{4\pi} \frac{h}{(r_c^{2d} + h^{2d})^{1/d}} \left[\frac{\mathbf{l}_{AB}}{|\mathbf{l}_{AB}|} \cdot \left(\frac{\mathbf{r}_{AP}}{|\mathbf{r}_{AP}|} - \frac{\mathbf{r}_{BP}}{|\mathbf{r}_{BP}|} \right) \right] \frac{\mathbf{r}_{AP} \times \mathbf{r}_{BP}}{|\mathbf{r}_{AP} \times \mathbf{r}_{BP}|}, \quad d = 2 \quad (2.1.11)$$

2.2. 自由渦後流・翼素理論による定式化

(1) ブレード形状と離散化

ブレード形状のねじり分布ピッチ角 β_t は、翼素の迎え角に影響を与える。また、ブレードをピッチ変更軸回りに回転させて設定する可変ピッチ角 $\pm |\beta_v|$ も、同様である。ブレードのピッチ角 β は、回転面を基準に、 β_t および β_v から、回転ブレード半径方向位置 r と半径 R の比を $r' = r/R$ として

$$\beta(r') = \beta_t(r') + \beta_v \quad (2.2.1)$$

になる。この β_t は、ピッチ p を導入し、直径 $D=2R$ からピッチ直径比 p/D として

$$p = 2\pi r \tan \beta_t \quad (2.2.2)$$

$$\frac{p}{D} = \frac{p}{2R} = \pi r' \tan \beta_t \quad (2.2.3)$$

と形式的に関係付けられる。それゆえ β_t は、式(2.2.4)で表せる。

$$\beta_t(r') = \tan^{-1} \left(\frac{p/D}{\pi r'} \right) \quad (2.2.4)$$

ブレードを象徴する揚力線は、近似してピッチ変更軸に置かれる束縛渦である。束縛渦の離散化は、ブレード半径方向 r/R に対して、翼幅を翼素に分割する節点を白丸、節点間の中点を翼素の代表点にして黒丸にとる。例えば、根元辺りの空力的に無効部分を除いたブレードを分割して、節点数 $J = 10$ に取れば代表点が分割数 $(J-1) = 9$ になり、そのようなブレード形状例が図 2.2.1 で表される。ここで図中にある表記は、代表点位置におけるブレードの翼弦長比 c/R 、ブレードの反り高さ比 t/R 、および代表点位置におけるブレードのねじり分布ピッチ角 BET_t (β_t) [deg.]である。加えて、 B はブレード枚数、NACA5315 はブレードの翼型名称、 j はブレード根から先端に向けた順番である。




図 2.2.1 ブレード形状の例

(2) 回転ブレードと座標

ブレード 1 回転の 1 周分割数 I から、分割ステップの回転角 ψ [rad.] が

$$\psi[\text{rad.}] = \frac{2\pi}{I} i, \quad i = 1 \dots MI \quad (2.2.5)$$

となる。ここで、 M はブレードの回転回数である。また、ブレードの回転数 n [Hz] から角速度 Ω [rad./s] は

$$\Omega[\text{rad./s}] = 2\pi n, \quad n[\text{Hz}] \quad (2.2.6)$$

である。それゆえ、時間 t および時間ステップ Δt は

$$t = \frac{\psi}{\Omega} = \frac{i}{nI} = i\Delta t, \quad i = 1 \dots MI \quad (2.2.7)$$

$$\Delta t = \frac{1}{nI} \quad (2.2.8)$$

と離散化される。代表点位置 r における回転接線方向の相対周速度 Ωr は

$$\Omega r = 2\pi n r [\text{m/s}] \quad (2.2.9)$$

となる。

ブレード枚数 B を 2 枚にして、手前に x 、右へ y 、および上へ z の正とする右手系直交座標を図 2.2.2 に描く。対気速度は、 x 軸成分で負の向きに V_x 、 z 軸成分で負の向きに V_z を定める。ブレードの回転面は、 y - z 面であり、チルト角 θ (回転軸が水平で 0° 、回転軸が垂直で 90°) をとる場合、 y - z 面を局所座標と

して、 y 軸を中心に反時計回りに正回り θ する。回転ブレードの回転軸方向へ流入する対気速度成分 V 、および接線方向から流入する対気速度成分 V_t は

$$\begin{pmatrix} V \\ V_t \end{pmatrix} = \begin{pmatrix} \cos(-\theta) & -\sin(-\theta) \\ \sin(-\theta) & \cos(-\theta) \end{pmatrix} \begin{pmatrix} V_x \\ V_z \end{pmatrix} \tag{2.2.10}$$

から変換して求める。図中で着目している束縛渦は、時間ステップが $i = 1$ ステップで Δt 経過して角度 ψ [rad.] の位置にくる。

基準となるブレードを象徴する束縛渦の初期 ($t = 0$) 位置は、 $\psi = 0^\circ$ とする。この初期位置は、 $i = MI$ で回り戻った最後の位置にもなる。 B 枚ブレードの他の束縛渦の初期位置は

$$\psi = \frac{360^\circ}{B}(k - 1), \quad k = 2 \dots B \tag{2.2.11}$$

と 360° を B で等分配する。数値計算において、他の離散化束縛渦およびそれから流出する自由渦の強さは、着目する束縛渦および自由渦の強さを用いる。また、他の自由渦後流系の形状は、着目する束縛渦の自由渦後流系の形状と同じにする。




図 2.2.2 座標と回転ブレードの束縛渦および自由渦 (2 枚ブレードでチルト角 $\theta=0^\circ$ と 90° の場合を併記)

図 2.2.2 の直交座標において、回転の変換は、ブレードの回転角 ψ に対し、 x 軸回りの回転行列 R_x として

$$R_x = \begin{pmatrix} 1 & 0 & 0 \\ 0 & \cos \psi & -\sin \psi \\ 0 & \sin \psi & \cos \psi \end{pmatrix} \tag{2.2.12}$$

を左から掛ける。さらに、ブレードの回転面がチルト角 θ をとる場合、 y 軸回りの回転行列 R_y として

$$R_y = \begin{pmatrix} \cos(-\theta) & 0 & \sin(-\theta) \\ 0 & 1 & 0 \\ -\sin(-\theta) & 0 & \cos(-\theta) \end{pmatrix} \tag{2.2.13}$$

を左から掛ける。

図 2.2.3 において、流入角 ϕ 、誘導迎角 α_i 、有効な合成速度 V_e とその軸方向成分 V_{w_a} および回転面接線方向成分 V_{w_t} とすれば、 V_e および回転面からの角度 $\phi + \alpha_i$ は

$$\begin{aligned} V_{w_t} &= \Omega r + V_t \sin \psi + w_t \\ V_{w_a} &= V + w_a \end{aligned} \quad (2.2.17)$$

$$V_e = \sqrt{V_{w_t}^2 + V_{w_a}^2} \quad (2.2.18)$$

$$\phi + \alpha_i = \tan^{-1} \frac{V_{w_a}}{V_{w_t}} \quad (2.2.19)$$

である。ブレードのピッチ角 β から、有効迎角 α_e は

$$\alpha_e = \beta - (\phi + \alpha_i) = \alpha - \alpha_i, \quad \alpha = \beta - \phi \quad (2.2.20)$$

になる。初期の状態は、誘導速度が分からないため、最初の時間ステップで $\alpha_e = \beta - \phi = \alpha$ を使う。

得られた α_e に対する翼型の与えられた揚力係数 $c_l(\alpha_e)$ から、 B 枚ブレードの単位翼幅当たりの翼力 L' は、飛行高度の大気密度 ρ および翼弦長 c を用い

$$\frac{L'}{B} = \frac{1}{2} \rho V_e^2 c(1) c_l \quad (2.2.21)$$

と表せる。一方、クッタ・ジューコフスキーの定理により

$$\frac{L'}{B} = \rho V_e \Gamma \quad (2.2.22)$$

の関係があるため、これらの式から、翼素断面の循環 Γ が、次式で求められる。

$$\Gamma = \frac{V_e c c_l}{2} \quad (2.2.23)$$

循環 Γ は、翼素幅で一定な束縛渦の強さになり、 j 番目の離散化束縛渦の強さは

$$\Gamma_j = \frac{V_{e,j} c_j c_{l,j}}{2}, \quad j = 1 \dots J - 1 \quad (2.2.24)$$

である。また、 Γ_j を持つ離散化束縛渦の接点 j から流れ出る自由渦の強さ $\Delta\Gamma_j$ は、1 から J まで

$$\begin{aligned} \Delta\Gamma_1 &= \Gamma_1 - 0.0 \\ \Delta\Gamma_j &= \Gamma_j - \Gamma_{j-1}, \quad j = 2 \dots J - 1 \\ \Delta\Gamma_J &= 0.0 - \Gamma_{J-1} \end{aligned} \quad (2.2.25)$$

と定める。

自由渦後流の強さが決まるため、着目する過去の自由渦後流の節点 P への誘導速度は、束縛渦の強さ Γ 、および自由渦後流の強さ $\Delta\Gamma$ を持ったそれぞれ節点間の直線のベクトル \mathbf{AB} から、ビオ・サバルの法則の式(2.1.11)により求めて、総和をとる。その際に、ブレード枚数 B に応じて、ある回転角をとる基準としたブレードの位置座標に対し、式(2.2.11)の初期位置を加えた位置にブレードがあると想定し、また対応する Γ および $\Delta\Gamma$ を同じにとり、これらブレードの過去の各節点間 \mathbf{AB} から点 P への誘導速度を計算する。

自由渦後流の接点位置の移動は、着目する節点 P における誘導速度の総和を $(w_x, w_y, w_z)_P$ として、時間ステップ Δt 経過後の新たな位置座標 $(x, y, z)_{P,NEW}$ を

$$(x, y, z)_{P,NEW} = (x, y, z)_{P,OLD} + (-V_x + w_x, w_y, -V_z + w_z)_P \Delta t \quad (2.2.26)$$

から与える。

空力特性値は、 B 枚ブレードの推力係数 C_T 、吸収するパワー係数 C_P 、および推進パワー（出力）と吸収パワー（入力）の比である推進効率 η は、 B 枚ブレードの推力 T および吸収するパワー P 、さらに軸方向対気速度 V として

$$C_T = \frac{T}{\rho n^2 D^4} \quad (2.2.27)$$

$$C_P = \frac{P}{\rho n^3 D^5} \quad (2.2.28)$$

$$\eta = \frac{VT}{P} \quad (2.2.29)$$

の形式を用いる。なお、ローターで用いられる推力係数 $C_{T,R}$ およびパワー係数 $C_{P,R}$ は、 $T = \rho \pi R^2 (\Omega R)^2 C_{T,R} = (\pi^3/4) \rho n^2 D^4 C_{T,R}$ から、 $C_T = (\pi^3/4) C_{T,R}$ の関係にあり、同様に、 $P = \rho \pi R^2 (\Omega R)^3 C_{P,R} = (\pi^4/4) \rho n^3 D^5 C_{P,R}$ から、 $C_P = (\pi^4/4) C_{P,R}$ の関係にある。両者の推力係数およびパワー係数は、係る定数が違うだけなので、相似則におけるパイナンバーのニュートン数⁹⁾に変わらない。

図 2.2.3 の翼素断面における力のベクトル図を参照して、局所推力 dT および局所パワー dP は、翼素の翼弦長 c で翼型の揚力係数 c_l および抗力係数 c_d が

$$dL = \frac{B}{2} \rho V_e^2 c d r c_l \quad (2.2.30)$$

$$dD = \frac{B}{2} \rho V_e^2 c d r c_d \quad (2.2.31)$$

であるから

$$dT = \frac{dL}{B} \cos(\phi + \alpha_i) - \frac{dD}{B} \sin(\phi + \alpha_i) = \frac{B}{2} \rho V_e^2 c d r \{c_l \cos(\phi + \alpha_i) - c_d \sin(\phi + \alpha_i)\} \quad (2.2.32)$$

$$dQ/r = \frac{dL}{B} \sin(\phi + \alpha_i) + \frac{dD}{B} \cos(\phi + \alpha_i) = \frac{B}{2} \rho V_e^2 c d r \{c_l \sin(\phi + \alpha_i) + c_d \cos(\phi + \alpha_i)\} \quad (2.2.33)$$

$$dP = 2\pi n dQ = \pi n B \rho V_e^2 c r d r \{c_l \sin(\phi + \alpha_i) + c_d \cos(\phi + \alpha_i)\} \quad (2.2.34)$$

と関係付けられる。式(2.2.32)および式(2.2.34)を、それぞれ式(2.2.27)及び(2.2.28)のように係数化すると

$$dC_T = \frac{dT}{\rho n^2 D^4} = \frac{B V_e^2 c' d r'}{32 n^2 R^2} c_d \left\{ \frac{c_l}{c_d} \cos(\phi + \alpha_i) - \sin(\phi + \alpha_i) \right\} \quad (2.2.35)$$

$$dC_P = \frac{dP}{\rho n^3 D^5} = \frac{\pi B V_e^2 c' r' d r'}{32 n^2 R^2} c_d \left\{ \frac{c_l}{c_d} \sin(\phi + \alpha_i) + \cos(\phi + \alpha_i) \right\} \quad (2.2.36)$$

になる。ここで、添字'付きの記号は、 R で除した無次元長さである。さらに、翼素剛率 (blade element solidity) $s = B c d r / (2\pi r d r)$ を導入して

$$s = B c' / (2\pi r') \quad (2.2.37)$$

から

$$dC_T = \frac{dT}{\rho n^2 D^4} = \frac{\pi s V_e^2 r'}{16 n^2 R^2} c_d \cos(\phi + \alpha_i) \left\{ \frac{c_l}{c_d} - \tan(\phi + \alpha_i) \right\} d r' \quad (2.2.38)$$

$$dC_P = \frac{dP}{\rho n^3 D^5} = \frac{\pi^2 s V_e^2 r'^2}{16 n^2 R^2} c_d \cos(\phi + \alpha_i) \left\{ \frac{c_l}{c_d} \tan(\phi + \alpha_i) + 1 \right\} d r' \quad (2.2.39)$$

と整理できる。そこで、 r' に対する特性分布は、 j 番目の翼素で

$$\left(\frac{dC_T}{dr'}\right)_j = \frac{\pi s V_e^2 r'}{16n^2 R^2} c_d \cos(\phi + \alpha_i) \left\{ \frac{c_l}{c_d} - \tan(\phi + \alpha_i) \right\} \quad (2.2.40)$$

$$\left(\frac{dC_P}{dr'}\right)_j = \frac{\pi^2 s V_e^2 r'^2}{16n^2 R^2} c_d \cos(\phi + \alpha_i) \left\{ \frac{c_l}{c_d} \tan(\phi + \alpha_i) + 1 \right\} \quad (2.2.41)$$

になる。これらの式へ翼素幅 $\Delta r'_j$ を掛け、総和を取れば

$$C_T = \sum_{j=1}^{J-1} \left(\frac{dC_T}{dr'}\right)_j \Delta r'_j \quad (2.2.42)$$

$$C_P = \sum_{j=1}^{J-1} \left(\frac{dC_P}{dr'}\right)_j \Delta r'_j \quad (2.2.43)$$

と、回転ブレード B 枚ブレードの推力係数 C_T およびパワー係数 C_P が得られる。さらに、これらの式から進行率 $J_V = V/(nD)$ として、式(2.2.29)の推進効率 η が式(2.2.44)で求められる。なお、進行率は流入角を決める因子であり、ブレード端の速度比 $\lambda = V/(\Omega R)$ を進行率とすれば、 $\lambda = V/(\pi n D)$ から、 $\lambda = J_V/\pi$ の関係にある。

$$\eta = J_V \frac{C_T}{C_P}, \quad J_V = \frac{V}{nD} \quad (2.2.44)$$

負推力の風車状態など負のパワー係数 ($C_P < 0$) が現れる状態の効率は、負推力が抗力を意味し式(2.2.44)の η で入力と出力が逆になる¹²⁾ため

$$\text{Efficiency on negative power}(C_P < 0) \text{ state: } \frac{1}{\eta} \text{ (inverse efficiency)} \quad (2.2.45)$$

と η の逆数により示すことができる^{13), 14)}。この効率 $1/\eta$ を、逆効率と呼ぶことにする。

さらに、特にヘリコプターのホバリング性能を示すフィギュア・オブ・メリット (figure of merit) FM は、ホバーに必要なローターの理想パワー P_i と実際のパワー P の比から、 $V = 0$ の静止推力状態における効率として

$$FM = \frac{P_i}{P} \quad (2.2.46)$$

で表される¹³⁾。 P_i は、運動量理論とベルヌーイの定理を適用し、作動円板の面積 A 、対気密度 ρ 、軸方向の誘導速度 w 、および推力 T から、 $V = 0$ を考慮して w を消去すれば

$$P_i = T(V + w) = Tw \quad (2.2.47)$$

$$T = \rho A (V + w) 2w = 2\rho A w^2 \quad (2.2.48)$$

$$P_i = T \sqrt{\frac{T}{2\rho A}} = \frac{T^{3/2}}{\sqrt{2\rho A}} \quad (2.2.49)$$

の関係式を得る。それゆえ、 FM は式(2.2.46)へ式(2.2.49)を代入し

$$FM = \frac{1}{\sqrt{2\rho(\pi D^2/4)}} \frac{T^{3/2}}{P} = \sqrt{\frac{2}{\pi}} \frac{C_T^{3/2}}{C_P}, \quad V = 0 \quad (2.2.50)$$

で示される。ここで係数は、 $T = \rho n^2 D^4 C_T$ および $P = \rho n^3 D^5 C_P$ で定めた、それぞれ式(2.2.27)の C_T および式(2.2.28)の C_P を使用している。

代表点における計算結果がある程度に収束する IM 回転の空力特性値が得られたなら、その最後の計算結果を空力特性値と定める。

(4) 翼型空力特性の近似

翼型の迎え角をとる有効迎角 α_e に対して、近似特性 $c_l(\alpha_e)$ および $c_d(\alpha_e)$ は、 $\pm|\alpha_e|$ でそれぞれ通常領域の小迎え角、 $\pm 90^\circ$ を超えるまでの大迎え角、およびその間の中迎え角に区分けする。

- ・ 正負の小迎え角の近似特性は、風洞試験などで得られている所望の翼型特性曲線から近似式を求める。
- ・ $\pm 90^\circ$ を超える大迎え角の近似特性は、失速後の翼型特性で一般化された近似式¹⁴⁾を、式(2.2.51)および式(2.2.52)のように用いる。ここで、 α_e [rad.]である。

$$c_l(\alpha_e) = 2.0 \sin \alpha_e \cos \alpha_e \quad (2.2.51)$$

$$c_d(\alpha_e) = 2.0 \sin^2 \alpha_e \quad (2.2.52)$$

- ・ 小迎え角と大迎え角に挟まれた正負の中迎え角で近似特性は、小迎え角近似特性の端と大迎え角近似特性の端を、直線などで適切に結ぶ。

3. 対象ブレードと数値計算結果

自由渦後流による翼素理論を用いた Scilab (フリーのソフトウェア) プログラムの数値計算は、付録に掲載したメイン・プログラム「FVW-BET_rot2.3.sce」を Scilab で実行している。実行中に読み込まれるデータは、翼型名を含むブレード形状を記した csv ファイルで、例えばファイル名「FVW_Blade_RAF6_B2e.csv」である。

メイン・プログラムに組み込まれるサブ・プログラムは、「FVW_clcd_WS.sce」としてに、翼型の c_l および c_d 特性の近似式を関数「cl_WS($\alpha_e, r', Airfoil$)」および「cd_WS($\alpha_e, r', Airfoil$)」に、あらかじめ与えて設定しておく。また、「FVW_atmos.sce」は、標準大気関数が入っており、指定した飛行高度に対する大気密度 ρ だけでなく、必要ならレイノルズ数 (Reynolds number) 計算用の動粘性係数やマッハ数 (Mach number) 計算用の音速が得られる。

数値計算における条件は、渦核の半径 r_c を $r_c=0.018$ m に設定した。さらに、「FVW_clcd_WS.sce」で与える翼型の空力特性は、臨界レイノルズ数以上¹³⁾を設定し、レイノルズ数効果の影響が小さい想定にした。

空力特性値の出力は、「FVW_」を冠する時系列名の csv ファイルが、フォルダ「Results」に作成される。さらに、図 3.1 のように、特性の半径方向分布や自由渦系の形状を出力ウィンドウに画面表示する。なお、プログラム構成や実行方法、またプログラム内容は、付録に掲載している。

計算結果の空力特性値は、1回転分割数 $I=12$ および回転回数 $M=2$ に取れば、図 3.2 に示すように特性値の有効数値 2桁程度の精度で収束する。なお、 $J_r=0$ の収束性で、CP 値が変化するものの僅であるから、 $M=2$ の空力特性値を用いる。ただし、3.1 (3) で言及するように、 $J_r=0$ において β_r の値により空力特性値が不安定になる場合、その FM 値は無効とする。

1点の進行率に対する計算速度は、さらに束縛渦の接点数 $J=11$ と 2枚ブレードの条件で、CPU の最大周波数 3.4GHz のパソコンを使用して約 2.5 min.かかる。

以下の空力特性グラフにおいて、表示される記号「ETA」および「1/ETA」はそれぞれ B 枚ブレードの推進効率 η および逆効率 $1/\eta$ 、記号の添え字「_exp.」はその記号の風試データ、「BETv」は可変ピッチ角 β_v 、「THE」はチルト角 θ 、「ALP」および「ALPe」はそれぞれ迎え角 α および有効迎え角 α_e 、ならびに「GAM」は循環 Γ を示す。




図 3.1 計算結果出力の画面表示ウィンドウ例
 (R.A.F.6、 $B=2$ 、 $\beta_v=0^\circ$ 、 $J=0.656$ 、 $n=2000/60$ Hz、 $V=20$ m/s)




図 3.2 空力特性値の収束性 (M: 回転回数)

3.1. R.A.F.6 翼型を持つ 3 feet プロペラ

本プログラムによる数値計算と比較する風洞試験¹⁷⁾は、駆動部内装の細長いナセルがプロペラ翼根を覆い、その細長いナセルを吊り線で支持して力やモーメントを計測する方式である。細長いナセルであるため、プロペラ前後の流れ干渉が小さくなる支持方式と考える。

風洞試験で用いられたブレードは、半径位置 $r/R = 0.75$ で R.A.F.6 翼型を持ち、直径 D が 3 フィートでブレード枚数 B が 2 枚あるいは 4 枚になる。ブレードの捩り分布ピッチ角 β_t は、ピッチ直径比 $p/D = 1.5$ 一定 ($\beta_t = 37.5^\circ @ r/R = 0.75$) である。風洞試験データ¹⁷⁾は、 $p/D = 1.5$ 一定とこれに可変ピッチ角 β_v として -9.9° および 10.7° を加えたブレード形態に対し、試験された通常状態の空力特性データを利用する。

R.A.F.6 翼型の空力特性は、文献のグラフ¹⁸⁾から引用して小迎え角領域の近似式を与える。風洞試験のブレードは、半径方向に翼型の翼厚が変化する。翼型の抗力係数 c_d は、半径方向位置で翼厚分の抗力係数増加を考慮している。しかし、翼型の揚力係数 c_l の計算は、半径方向に R.A.F.6 翼型だけを設定している。空力特性の概算に用いたブレード形状を、R.A.F.6 翼型の形状と併せて図 3.1.1 に表す。




図 3.1.1 R.A.F.6 翼型形状とブレード形状 (2 枚および 4 枚ブレードともに同形)

(1) 2 枚ブレードの計算結果と風試データの比較

2 枚ブレードによる空力特性の計算結果と風試データの比較を、図 3.1.2～図 3.1.4 に示す。図 3.1.2




図 3.1.2 2 枚ブレード計算結果と風試データの比較 (右図：部分拡大) ピッチ直径比 1.5




図 3.1.3 2枚ブレード計算結果と風試データの比較；ピッチ直径比 1.5 可変ピッチ角 -9.9°




図 3.1.4 2枚ブレード計算結果と風試データの比較；ピッチ直径比 1.5 可変ピッチ角 10.7°

の計算結果と風試データの比較において、左図は特性の全体を示し、特性曲線で進行率 J_v が小さく（翼素の迎え角は大きく）なるに従い $J_v = 1.8$ 辺りから立ち上がり、推進効率 η が最大値を過ぎた後 $J_v = 1.0$ 程度までの領域で、計算結果は風試データとほぼ良い一致を示している。この領域で各翼素の有効迎え角は小さく、翼型の線形的な特性領域にあると考える。

$J_v = 1.0$ 以下の領域は右図に拡大してあり、計算結果は $J_v = 0.3$ まで滑らかな特性が続くものの、風試データとの差が最大で 20%程度現れる。この領域で各翼素の有効迎え角は、中程度になり、翼型の失速特性前後を含む領域が該当すると考える。

進行率を小さくして $J_v = 0.164 \sim 0.033$ の領域で計算結果の特性は、不連続を示す。この領域は、右図に添付してある計算結果で自由渦後流系の様子から分かるように、 $J_v = 0$ ($V = 0$) で一部の後流が前方へ流れており、途中 $J_v = 0.098$ ($V = 3$ m/s) で前方へ流れた後流が V で押し返されきつ抗し、さらに $J_v = 0.197$ ($V = 6$ m/s) に進むと後流全体が V に流されている。それゆえ、対気速度と誘導速度の大きさが同程度となって回転面周辺に自由渦後流が集り、それらの後流による複雑な誘導速度が生じ、空力特性が J_v の小さな変化で急変する不安定な領域になると推測する。

$J_v = 0.033 \sim 0.0$ の領域で計算結果は、おおよそ連続している。この領域で各翼素の有効迎え角は大きく、誘導速度だけが特性に影響を与えていると考える。

図 3.1.3 および図 3.1.4 の特性比較図は、図 3.1.2 特性のブレードで、可変ピッチ角 β_v をそれぞれ -9.9° および 10.7° に取った特性曲線である。 $\beta_v = -9.9^\circ$ の特性比較は、図 3.1.2 の特性より全体に小さく、 $J_v = 1.2 \sim 0.3$ の領域で計算結果と風試データが全体の割合から見て広い範囲でほぼ一致する。図 3.1.4 において、 $\beta_v = 10.7^\circ$ の特性比較は、図 3.1.2 の特性より J_v が 1 程度加わった範囲まで広がり、特性が大きくなって、計算結果と風試データの特性差が最大 28%程度に拡大されている。 $\beta_v = 10.7^\circ$ の流入角は $\beta_v = 0^\circ$ の流入角よりも 10.7° 大きくなり、 $\beta_v = 0^\circ$ の空力特性の誤差が、流入角の増加分拡大されたと考えられる。特性曲線の大きな変曲点は、図 3.1.2、図 3.1.3、および図 3.1.4 共に、 $J_v = 0.15$ 辺りになっている。




図 3.1.5 4枚ブレード計算結果と風試データの比較 (右図：部分拡大) ピッチ直径比 1.5

(2) 4枚ブレードの計算結果と風洞試験データの比較

4枚ブレードによる空力特性の計算結果と風試データの比較を、図 3.1.5 および図 3.1.7・8 に示す。図中グラフの $C_T \cdot C_P$ 軸は、2枚ブレードの2倍の数値が振られている。図 3.1.5 の左図で、 $J_V = 1.8 \sim 1.1$ 程度まで計算結果と風試データはほぼ一致し、2枚ブレードの図 3.1.2 左図と同一 J_V に対するブレード1枚当たりの C_T 特性値は、4枚ブレードのほうが小さくなっている。これは、ブレード枚数増加による干渉の影響が、計算結果に表されていると考える。図 3.1.5 の右図で、 $J_V = 0.5$ 程度より小さくなると、計算結果は特性の振れおよび風試データとの差が大きくなる。

ブレード枚数による空力特性値への影響を調べるため、ブレード枚数 $B=2$ および 4 の計算結果が、風試データとほぼ一致する特性曲線の立ち上がり傾斜領域で、 C_T の比較的大きい進行率 $J_V=1.31$ を取り上げる。この $J_V = 1.31$ の設定で、 B を 1 から 6 まで増加した計算結果の空力特性値が、図 3.1.6 である。図中の破線直線は、 $B = 1$ の特性値を基準にした B 倍の特性値になる。 B の増加に従い、 C_T および C_P 値は増加するも、破線直線の基本特性倍数値より小さく、それとの開きが増加する。加えて、推進効率 η は、 $B=2$ から 6 への増加に従い単調に減少し、また $B = 2$ より 1 のほうが僅かに大きい。これらは、ブレード枚数を伴う空力特性値の特徴そのものとする。




図 3.1.6 ブレード枚数に対する空力特性値：進行率 $J_V=1.31$ ($n=2,000/60$ Hz, $V=40$ m/s)

β_v を -9.9° および 10.7° に設定した特性比較が、それぞれ図 3.1.7 および図 3.1.8 であり、 $\beta_v = 10.7^\circ$ の特性比較は、 $\beta_v = 0^\circ$ である図 3.1.5 の比較差が拡大されている。特性曲線の大きな変曲点は、図 3.1.5、図 3.1.7、および図 3.1.8 共に、 $J_v = 0.55 \sim 0.5$ 辺りになる。




図 3.1.7 4枚ブレード計算結果と風試データの比較；ピッチ直径比 1.5 可変ピッチ角 -9.9°




図 3.1.8 4枚ブレード計算結果と風試データの比較；ピッチ直径比 1.5 可変ピッチ角 10.7°

(3) 2枚ブレードの可変ピッチ角変化による全状態計算結果

2枚ブレードの可変ピッチ角 β_v 変化による全状態空力特性の計算結果を、図 3.1.9～図 3.1.13 に示す。この 2枚ブレードの全状態空力特性を調べるため、(1)項で示した比較特性の計算結果特性を基にして、より大きい J_v 領域へ範囲を拡大する。さらに、 β_v を負側へ -20° 、 -30° 、および -40° に設定して加え、




図 3.1.9 2枚ブレード可変ピッチ角変化の全状態計算結果：推力係数特性




図 3.1.10 2枚ブレード可変ピッチ角変化の全状態計算結果：パワー係数特性




図 3.1.11 2枚ブレード可変ピッチ角変化の全状態計算結果：推進効率特性




図 3.1.12 2枚ブレード可変ピッチ角変化の全状態計算結果：逆効率特性

逆向きの対気速度になる負の J_V 領域にまで拡張して、空力特性を計算している。

β_v 変化による推力係数 C_T の特性曲線が、図 3.1.9 になる。どの β_v の特性曲線も負推力係数が表され、特性曲線の傾向として、 J_V に対して負の勾配を示す。特性曲線で $J_V = 0$ 前後に現れる変動は、(1)項で触れたように、対気速度と誘導速度の干渉により不安定な特性になっていると考える。これら C_T の特性曲線に対応するパワー係数 C_P の特性曲線が、図 3.1.10 である。 $\beta_v = -40^\circ$ を除く特性曲線は、正の J_V 領域（正の対気速度）で負推力係数の領域に付き負パワー係数の領域があり、負推力の風車状態を示す。一方、 $\beta_v = -30^\circ$ および -40° の特性曲線は、負の J_V 領域（負の対気速度）において負パワー係数を示す領域がある。この領域の状態は、対応する推力係数が正であることから正推力の風車状態と呼ぶことにする。なお、正・負推力の風車状態にある風車は、飛行機上の風車^{13), 14)} 状態であり、飛行実証で実施したプロペラ回生¹⁹⁾ は、負推力の風車状態になる。

図 3.1.11 は、式(2.2.44)で表される推進効率 η の特性曲線である。負のパワー係数であり、正の J_V 領域にある負推力の風車状態および負の J_V 領域にある正推力の風車状態の効率が、式(2.2.45)で表される逆効率 $1/\eta$ の特性曲線として図 3.1.12 に示される。なお、グローアート^{13), 14)} によれば、飛行機上の風車の効率は、運動量理論から、 $2/3 = 0.667$ のときに出力（負の吸収パワー）が最大になる。




図 3.1.13 2枚ブレードの可変ピッチ角に対する静止推力状態の計算結果：推力係数特性を伴うフィギュア・オブ・メリット特性

軸方向対気速度 V が 0 の静止推力状態にある効率は、式(2.2.50)で表されるフィギュア・オブ・メリット FM であり、図 3.1.13 に C_T を伴い可変ピッチ角 β_v に対する特性曲線を示す。 $\beta_v = -20^\circ$ 以下の FM 値は、数値計算が不安定で無効な値になるため、グラフに示していない。静止推力状態において、このブレードは C_T の値が許容できれば $\beta_v = -9.9^\circ$ で FM 値約 0.45 を得る。

4. まとめ

プロペラの回転ブレードの全状態空力特性を簡便に得るため、自由渦後流・翼素理論による数値計算プログラムを Scilab で作成して実行した。プロペラの通常状態を使用し、空力特性の計算結果と風試データとを比較した。その結果、プログラムによる計算結果は、総じて、空力特性の特徴をつかめる概算になった。さらに計算の領域を、空力特性の静止推力状態を介し、主眼である負推力の風車状態および正推力の風車状態まで延長して、空力特性曲線を取得できた。

参考文献

- 1) Barnes W. McCormick: Aerodynamics Aeronautics and Flight Mechanics, John Wiley & Sons, 1995, pp.299-310.
- 2) L. Prandtl (Translated by the Staff of the NACA): Applications of Modern Hydrodynamics to Aeronautics, NACA-TR-116, 1923.
- 3) 原田正志: 3B08 プロペラ設計法の精度, 第 51 回飛行機シンポジウム JSASS-2013-5155, 2013.11.20-22.
- 4) 原田正志: 静止推力を最大とするダクテッドファン設計法, 日本航空宇宙学会論文集 Vol. 66, No.3, 2018, pp.75-83.
- 5) S. Hauptmann, M. Bülk, L. Schön, S. Erbslöh, K. Boorsma, F. Grasso, M. Kühn, P. W. Cheng: Comparison of the lifting-line free vortex wake method and the blade-element-momentum theory regarding the simulated loads of multi-MW wind turbines, Journal of Physics: Conference Series 555 (2014) 012050.
- 6) A. van Garrel: Development of a wind turbine aerodynamics simulation module, TR ECN-C-03-079, ECN, 2003.
- 7) 西村哲夫: 流体力学 (I), 日本工業新聞社, 1971, pp.86-88.
- 8) Leishman, J. G.: Principles of Helicopter Aerodynamics, Cambridge University Press, 2006, pp.599-601.
- 9) Vastias, G. H., Kozel, V., and Mih, W. C.: A Simpler Model for Concentrated Vortices, Experiments in Fluids, 11, 1991, pp.73-76.
- 10) Leishman, J. G.: Principles of Helicopter Aerodynamics, Cambridge University Press, 2006, pp.586-592.
- 11) 奥山政広, 小林宙, 西沢啓: プロペラにおける風車状態の空力特性, JAXA-RM-14-010, 2015, 付録.
- 12) 奥山政広, 小林宙, 西沢啓: プロペラにおける風車状態の空力特性, JAXA-RM-14-010, 2015.
- 13) H. Glauert: The elements of aerofoil and airscrew theory, Cambridge University Press, 1947(1993), pp.206-207.
- 14) エイチ・グローアト (内藤孟 訳): 翼及プロペラ理論, 雄山閣, 1943, pp.201-202.
- 15) Barnes W. McCormick: Aerodynamics of V/STOL Flight, Dover Publications, Inc., 1967(1999), pp.106-109.
- 16) S. F. Hoerner, and H. V. Borst, Fluid-Dynamic Lift, 1985, pp.(4-22)-(4-23).
- 17) C. N. H. Lock, H. Bateman, and H. L. Nixon, Wind Tunnel Tests of High Pitch Airscrews (Part 1), R. & M. No. 1673, 1934.
- 18) Fred E. Weick, Aircraft Propeller Design, McGraw-Hill, 1930, p.34.
- 19) 西沢啓, 小林宙, 飯島朋子, 他: 航空機用電動推進システム技術の飛行実証, JAXA-RM-16-006, 2017.

付 録

A.1. プログラム構成

(A1) Scilab 作業フォルダ内のディレクトリ

Results (計算出力のフォルダ) → FVW_202109151229_I12M2N2000BET10.7THE0V0Vt0.csv (計算出力のファイル例)

FVW-BET_rot2.3.sce (メイン・プログラム)

FVW_clcd_WS.sce (サブ・プログラム：与える翼型空力特性の c_l および c_d 近似式)

FVW_atmos.sce (サブ・プログラム：標準大気で飛行高度の大気密度などが得られる)

FVW_Blade_RAF6_B4e.csv (与えるブレード形状データと翼型名のファイル例)

FVW_clcd_WS_plot.sce (与えた翼型空力特性のプロット用確認プログラム)


(A2) 事前準備と実行

- ・与えるブレード形状ファイルを作成し、また「FVW_clcd_WS.sce」に与える翼型特性近似式を追加する。
- ・記憶装置内 (Scilab を実行するパソコンの「ドキュメント」フォルダ内も含めて) に例えば「FVW-BET」名のフォルダを作成し、(A1)項のフォルダやファイルを置く。Scilab をインストールしたパソコンで、「FVW-BET」フォルダ内の「FVW-BET_rot2.3.sce」をダブルクリックすれば、Scilab が起動して「ファイルブラウザ」ウィンドウに「FVW-BET」フォルダのディレクトリが表示される。また、「SciNotes」ウィンドウにプログラムが表示される。
- ・「SciNotes」ウィンドウで、「実行」(あるいはプログラムを修正した場合、「保存して実行」)をクリックすれば、プログラムの実行が始まり「コンソール」ウィンドウに状況が表示される。
- ・ダイアログで、読み込む形状ファイルのデフォルト名が表示される。



必要なら修正して「OK」をクリック。
(デフォルト名の変更は、FVW-BET_rot2.2.sce で [x_dialog](#) 内を修正)

- ・ブレード形状がグラフィックス・ウィンドウで表示される。「ファイル」－「保存」で保存できる。
- ・マルチダイアログで、パラメータ値のデフォルトが表示される。



必要なら修正して「OK」をクリック。
(デフォルト値の変更は、FVW-BET_rot2.2.sce で [x_mdialog](#) の Def 内を修正)

- ・計算結果の値や計算時間がディスプレイに表示され、「Results」フォルダ内に出力データファイルが作成される。
- ・計算結果が、グラフィックス・ウィンドウで表示される。「ファイル」－「保存」で保存できる。

A.2. メイン・プログラム (FVW-BET_rot2.3.sce)

```
// Blade element theory by free vortex wake for estimating aerodynamic characteristics of rotatory blades
// FVW-BET_rot2.3.sce, 2022.1.26
// by OKUYAMA Masahiro
// on Scilab 6.1.0
//
//===== main characters =====
//a_rho(H) function of air density on the H
//alp(ALP) angle of attack about bare inflow
//alpe(ALPe) effective angle of attack
//B number of blades
//bett twisted pitch angles
```

```

//betv variable pitch angles
//bcpx x-distance at control points on the bound vortex
//bcpz z-distance at control points on the bound vortex
//bnpx x-distance at node points on the bound vortex and free vortexes
//bnpz z-distance at node points on the bound vortex and free vortexes
//chord chord length of the blade
//CT thrust coefficient
//CP power coefficient
//D diameter of the rotating blade, m
//eta(ETA) efficiency as propeller
//FM figure of merit, efficiency on state of hovering
//gam(GAM) circulation, m^2/s
//H flight altitude, m
//I division number of a rotation
//ieta(1/ETA) efficiency for regenerative power(negative torque) region of aerodynamic characteristics
//J division number of the blade span
//Jv Jv=V/(nD), advance ration
//M number of rotations
//N rotating speed of the blade, rpm
//n rotating speed of the blade, Hz(rps)
//ome ome=2.0*pi*n, angular velocity, rad/s
//P power
//psi angle of rotation(argument of x axes around z axes)
//R R=D/2, radius of the rotating blade(span of one blade), m
//Ry rotation matrix around y axes
//Rx rotation matrix around x axes
//r distance from the axis of rotation along the horizontal projection of the blade
//rc vortex core radius, m
//T thrust
//THE tilt angle within verti.:~ 90 deg. ~ hori.:0 deg. ~
//THEy argument of z axes around y axes
//V axially relative airspeed, m/s
//t height of the vertical projection of the blade
//Vt tangentially relative airspeed, m/s
//Vx horizontally relative airspeed, m/s
//Vz vertically relative airspeed, m/s
//wxyz induced velocity vector, m/s
//
//===== define functions and load sub-program files =====
function rad=d2r(deg)
  rad=deg/180*pi
endfunction
function deg=r2d(rad)
  deg=rad/pi*180
endfunction
//
exec("FVW_clcd_WS.sce");
exec("FVW_atmos.sce");
funcprot(0);
//
function v=v_mag(X)
  v=sqrt(X(1)^2+X(2)^2+X(3)^2)
endfunction
function w=BioSav(pa, pb, pp, g, rc)
  lab=pb-pa
  rap=pp-pa
  rbp=pp-pb
  varap=v_mag(rap)
  varbp=v_mag(rbp)
  vcraprbp=cross(rap,rbp)
  vacraprbp=v_mag(vcraprbp)
//  w=g/(4.0*pi)*(lab*(rap/varap-rbp/varbp))*vcraprbp/vacraprbp^2
  valab=v_mag(lab)
  hp=vacraprbp/valab
  w=g/(4.0*pi)*hp/((rc^4+hp^4)^0.5)*((lab/valab)*(rap/varap-rbp/varbp))*vcraprbp/vacraprbp
endfunction
//
//===== read the data file of the blade =====
//

```

```

rDfn=x_dialog("reading data file name","FVW_Blade_RAF6_B2e.csv") // <== make an adjustment as needed
//
Frd=mopen(rDfn,"r")
L1=mgetl(Frd,1);
R=msscanf(1,mgetl(Frd,1,"%f");//length of a blade, m
D=2.0*R;
B=msscanf(1,mgetl(Frd,1,"%f");//number of blades
J=msscanf(1,mgetl(Frd,1,"%f");//division number of the blade span
Jm1=J-1;
FoJm1="%f,"
for j=1:Jm1 FoJm1=FoJm1+"%f," end
FoJ=JoJm1+"%f,"
bnpr=msscanf(mgetl(Frd,1),FoJ);//, npr/R
bnpt=msscanf(mgetl(Frd,1),FoJ);//, npt/R //
bcpr=msscanf(mgetl(Frd,1),FoJm1);//, cpr/R
bcpt=msscanf(mgetl(Frd,1),FoJm1);//, cpt/R //
chord=msscanf(mgetl(Frd,1),FoJm1);//, c/R
bett=msscanf(mgetl(Frd,1),FoJm1);//twisted pitch angles
bett=d2r(bett) //deg -> rad
WS=mgetl(Frd,1);// airfoils
WS=mtlb_strrep(WS,"","")
mclose(Frd);
bnpz=bnpr //
bnpx=bnpt //
bnpy(1:J)=0.0 //
bcpz=bcpr //
bcpx=bcpt //
bcpy(1:Jm1)=0.0 //
//
//===== display the shape of the blade =====
f1=scf()
set(f1,"figure_size",[600,600])
subplot(3,1,1); plot2d(bcpr,0.446*chord,2,rect=[0 -0.2 1 0.15]) //
plot2d(bcpr,-0.554*chord,2) //
plot2d(bcpr,0.196*chord,-4) //plot2d(bcpr,bcpy,-4)
gca().font_size = 2.0
title("Blade configuration, R="+string(R)+"m, B="+string(B)+"", "+WS); title fontsize 2.0
ylabel("c/R"); ylabel fontsize 2.0; xlabel("r/R"); xlabel fontsize 2.0; xgrid(1)
//
subplot(3,1,2); plot2d(bcpr,bcpt,-4,rect=[0 -0.05 1 0.2]);
plot2d(bnpr,bnpt,-9); gca().font_size = 2.0
plot2d(bnpr,bnpt,2); gca().font_size = 2.0
ylabel("t/R"); ylabel fontsize 2.0; xlabel("r/R"); xlabel fontsize 2.0
xgrid(1); legend("control points","node points",2)
//
subplot(3,1,3); plot2d(bcpr,r2d(bett),2,rect=[0 0 1 60])
plot2d(bcpr,r2d(bett),-4)
gca().font_size = 2.0
ylabel("BETt [deg]"); ylabel fontsize 2.0; xlabel("r/R"); xlabel fontsize 2.0; xgrid(1)
//
//===== input Parameters =====
I=12 //division number of a rotation
M=2 //number of rotations
betv=d2r(0) //variable pitch angle
N=2000 //rpm of the blade
Vx=0 //horizontally relative airspeed
Vz=0 //vertically relative airspeed
THE=d2r(0) //tilt angle with hori.:0deg and verti.:90deg
H=0 //flight altitude, m
//
Title="enter parametric values"
//
Prompt=["I division number of a rotation";"M number of rotations";...
"BETv[deg] variable pitch angles";"N[rpm] rpm of the blade";...
"Vx[m/s] horizontally relative airspeed";"Vz[m/s] vertically relative airspeed";...
"THE[deg] tilt angle with hori.:0deg and verti.:90deg";"H[m] flight altitude"]
Def=["12";"2";"0";"2000";"0";"0";"0";"0"] // <== make an adjustment as needed
//
Answer=x_mdialog(Title,Prompt,Def)
ans=msscanf(8,Answer,"%f")

```

```

I=ans(1)
M=ans(2)
betv=d2r(ans(3))
N=ans(4)
Vx=ans(5)
Vz=ans(6)
THE=d2r(ans(7))
H=ans(8)
//
//===== setting =====
tic()
//
bet=betv+bett // bet0=betv+bett0
//
THEy=-THE
cost=cos(THEy); sint=sin(THEy) //
//
R2=[cost -sint;sint cost] //
Vta=R2*[Vx;Vz]
V=Vta(1) //axially relative airspeed, m/s
Vt=Vta(2) //tangentially relative airspeed, m/s
//
Jm1=J-1
IM=I*M
IMp1=IM+1
n=N/60.0 //Hz(rps)
Jv=V/(n*D) //advance ration
ome=2*pi*n //angular velocity, rad/s
dpsi=2*pi/I
dt=1/(n*I)
db=2*pi/B
dnpr=bnpr(2:J)-bnpr(1:Jm1)
dnpt=bnpt(2:J)-bnpt(1:Jm1)
//
//===== initialization =====
npx=zeros(IMp1,J)
npy=zeros(IMp1,J)
npz=zeros(IMp1,J)
cpx=zeros(IMp1,Jm1)
cpy=zeros(IMp1,Jm1)
cpz=zeros(IMp1,Jm1)
nkx=zeros(IMp1,J,B)
nky=zeros(IMp1,J,B)
nkz=zeros(IMp1,J,B)
ckx=zeros(IMp1,Jm1,B)
cky=zeros(IMp1,Jm1,B)
ckz=zeros(IMp1,Jm1,B)
wx(1:Jm1)=0.0
wy(1:Jm1)=0.0
wz(1:Jm1)=0.0
wt(1:Jm1)=0.0
wa(1:Jm1)=0.0
gam=zeros(IMp1,Jm1)
ve=zeros(IMp1,Jm1)
alpp=zeros(IMp1,Jm1)
alp=zeros(IMp1,Jm1)
alppi=zeros(IMp1,Jm1)
alpe=zeros(IMp1,Jm1)
dgam=zeros(IMp1,J)
dCT(1:Jm1)=0.0
dCP(1:Jm1)=0.0
dCF(1:Jm1)=0.0
//
a=[0.0;0.0;0.0]
ak=[0.0;0.0;0.0]
b=[0.0;0.0;0.0]
bk=[0.0;0.0;0.0]
p=[0.0;0.0;0.0]
wxyz=[0.0;0.0;0.0]

```

```

wxyzc=[0.0;0.0;0.0]
//
Ry=[cost 0 sint;0 1 0;-sint 0 cost] //
//
psi=0
cosp=cos(psi); sinp=sin(psi) //
Rx=[1 0 0;0 cosp -sinp;0 sinp cosp]
for j=1:J
  np1=[bnpx(j);bnpy(j);bnpz(j)]*R ///
  npv=Ry*Rx*np1
  npx(1,j)=npv(1); npy(1,j)=npv(2); npz(1,j)=npv(3)
  for k=2:B
    psidk=psi+db*(k-1)
    cosk=cos(psidk); sink=sin(psidk)
    Rxb=[1 0 0;0 cosk -sink;0 sink cosk]
    nkx=Ry*Rxb*np1
    nkx(1,j,k)=nkx(1); nky(1,j,k)=nkx(2); nkz(1,j,k)=nkx(3)
  end
end
for j=1:Jm1
  cp1=[bcpx(j);bcpy(j);bcpz(j)]*R ///
  cpv=Ry*Rx*cp1 //
  cpk(1,j)=cpv(1); cpy(1,j)=cpv(2); cpz(1,j)=cpv(3)
  for k=2:B
    psidk=psi+db*(k-1)
    cosk=cos(psidk); sink=sin(psidk)
    Rxb=[1 0 0;0 cosk -sink;0 sink cosk]
    ckv=Ry*Rxb*cp1
    ckx(1,j,k)=ckv(1); cky(1,j,k)=ckv(2); ckz(1,j,k)=ckv(3)
  end
end
//
for j=1:Jm1
  omer=ome*bcpr(j)*R
  wx(j)=0; wy(j)=0; wz(j)=0
  wcy1=Ry.*[wx(j);wy(j);wz(j)] // Ry.' transposed matrix of Ry
  wn=wcyl(1); wr=wcyl(2); wpsi=wcyl(3)
  wt(j)=-wpsi
  wa(j)=-wn
  vwt=sinp*Vt+omer+wt(j) //
  vwa=V+wa(j)
  ve(1,j)=sqrt(vwt^2+vwa^2)
  alpp(1,j)=atan(V/omer) //
  alp(1,j)=bet(j)-alpp(1,j) //
  alppi(1,j)=atan(vwa/vwt)
  alpe(1,j)=bet(j)-alppi(1,j)
  gam(1,j)=ve(1,j)*chord(j)*R*c_l_WS(r2d(alpe(1,j)),bcpr(j),WS)/2 //
end
dgam(1,1)=gam(1,1)-0.0
for j=2:Jm1
  dgam(1,j)=gam(1,j)-gam(1,j-1)
end
dgam(1,J)=0.0-gam(1,Jm1)
//
rc=0.018 // vortex core radius, m <== make an adjustment as needed
//
//===== main calculations =====
for i=1:IM // time=i*dt=i/(nl)
  psi=i*dpsi //angle ofrotation
  cosp=cos(psi); sinp=sin(psi) //
  Rx=[1 0 0;0 cosp -sinp;0 sinp cosp]
  for j=1:J
    np1=[bnpx(j);bnpy(j);bnpz(j)]*R ///
    npv=Ry*Rx*np1
    npk(i+1,j)=npv(1); npy(i+1,j)=npv(2); npz(i+1,j)=npv(3)
    for k=2:B
      psidk=psi+db*(k-1)
      cosk=cos(psidk); sink=sin(psidk)
      Rxb=[1 0 0;0 cosk -sink;0 sink cosk]
      nkx=Ry*Rxb*np1
    end
  end
end

```

```

    nkx(i+1,j,k)=nkv(1); nky(i+1,j,k)=nkz(2); nkz(i+1,j,k)=nkv(3)
end
end
for j=1:m1
cp1=[bcpx(j);bcpy(j);bcpz(j)]*R ///
cpv=Ry*Rx*cp1
cpx(i+1,j)=cpv(1); cpy(i+1,j)=cpv(2); cpz(i+1,j)=cpv(3)
for k=2:B
psidk=psi+db*(k-1)
cosk=cos(psidk); sink=sin(psidk)
Rxb=[1 0 0;0 cosk -sink;0 sink cosk]
ckv=Ry*Rxb*cp1
ckx(i+1,j,k)=ckv(1); cky(i+1,j,k)=ckv(2); ckz(i+1,j,k)=ckv(3)
end
end
for i2=1:i
for j2=1:j
p=[npx(i2,j2);npy(i2,j2);npz(i2,j2)]*R ///
wxyz=[0.0;0.0;0.0]
for j3=1:j
for i3=1:i
if (i3~=i2)&(j3~=j2)
a=[npx(i3,j3);npy(i3,j3);npz(i3,j3)]*R ///
b=[npx(i3+1,j3);npy(i3+1,j3);npz(i3+1,j3)]*R /// !
wxyz=wxyz+BioSav(a,b,p,dgam(i,j3),rc)
end
for k=2:B
ak=[nkx(i3,j3,k);nky(i3,j3,k);nkz(i3,j3,k)]
bk=[nkx(i3+1,j3,k);nky(i3+1,j3,k);nkz(i3+1,j3,k)]
wxyz=wxyz+BioSav(ak,bk,p,dgam(i,j3),rc)
end
end
end
for j3=1:m1
a=[npx(i+1,j3);npy(i+1,j3);npz(i+1,j3)]*R ///
b=[npx(i+1,j3+1);npy(i+1,j3+1);npz(i+1,j3+1)]*R ///
wxyz=wxyz+BioSav(a,b,p,gam(i,j3),rc)
for k=2:B
ak=[nkx(i3+1,j3,k);nky(i3+1,j3,k);nkz(i3+1,j3,k)]
bk=[nkx(i3+1,j3+1,k);nky(i3+1,j3+1,k);nkz(i3+1,j3+1,k)]
wxyz=wxyz+BioSav(ak,bk,p,gam(i,j3),rc)
end
end
npx(i2,j2)=npx(i2,j2)+(-Vx+wxyz(1))*dt
npy(i2,j2)=npy(i2,j2)+wxyz(2)*dt
npz(i2,j2)=npz(i2,j2)+(-Vz+wxyz(3))*dt
for k=2:B
nkx(i2,j2,k)=nkx(i2,j2,k)+(-Vx+wxyz(1))*dt
nky(i2,j2,k)=nky(i2,j2,k)+wxyz(2)*dt
nkz(i2,j2,k)=nkz(i2,j2,k)+(-Vz+wxyz(3))*dt
end
end
end
//
for j=1:m1
p=[cpx(i+1,j);cpy(i+1,j);cpz(i+1,j)]*R ///
wxyzc=[0.0;0.0;0.0]
for j3=1:j
for i3=1:i
a=[npx(i3,j3);npy(i3,j3);npz(i3,j3)]*R ///
b=[npx(i3+1,j3);npy(i3+1,j3);npz(i3+1,j3)]*R ///
wxyzc=wxyzc+BioSav(a,b,p,dgam(i,j3),rc)
for k=2:B
ak=[nkx(i3,j3,k);nky(i3,j3,k);nkz(i3,j3,k)]
bk=[nkx(i3+1,j3,k);nky(i3+1,j3,k);nkz(i3+1,j3,k)]
wxyzc=wxyzc+BioSav(ak,bk,p,dgam(i,j3),rc)
end
end
end
end

```

```

omer=ome*bcpr(j)*R
wx(j)=wxyzc(1); wy(j)=wxyzc(2); wz(j)=wxyzc(3)
wcyl=Ry.*[wx(j);wy(j);wz(j)] // Ry.'
wn=wcyl(1); wr=wcyl(2); wpsi=wcyl(3)
wt(j)=-wpsi
wa(j)=-wn
vwt=sinp*Vt+omer+wt(j) //
vwa=V+wa(j)
ve(i+1,j)=sqrt(vwt^2+vwa^2)
alpp(i+1,j)=atan(V/omer) //
alp(i+1,j)=bet(j)-alpp(i+1,j)
alppi(i+1,j)=atan(vwa/vwt)
alpe(i+1,j)=bet(j)-alppi(i+1,j)
gam(i+1,j)=ve(i+1,j)*chord(j)*R*cl_WS(r2d(alpe(i+1,j)),bcpr(j),WS)/2 //
end
dgam(i+1,1)=gam(i+1,1)-0.0
for j2=2:Jm1
    dgam(i+1,j2)=gam(i+1,j2)-gam(i+1,j2-1)
end
dgam(i+1,J)=0.0-gam(i+1,Jm1)
end
//
for i=1:IM
    CT2(i)=0.0
    CP2(i)=0.0
    CF2(i)=0.0 //
    for j=1:Jm1
        s=B*chord(j)/(2*pi*bcpr(j))
        clj=cl_WS(r2d(alpe(i+1,j)),bcpr(j),WS) //
        cdj=cd_WS(r2d(alpe(i+1,j)),bcpr(j),WS) //
        cospi=cos(alppi(i+1,j)); tanpi=tan(alppi(i+1,j))
        dCT(j)=pi*s*bcpr(j)*ve(i+1,j)^2/(16.0*n^2*R^2)*cdj*cospi*(clj/cdj-tanpi)*dnpr(j)
        dCP(j)=pi*s*bcpr(j)^2*ve(i+1,j)^2/(16.0*n^2*R^2)*cdj*cospi*(clj/cdj*tanpi+1.0)*dnpr(j)
        dCF(j)=pi*s*bcpr(j)*ve(i+1,j)^2/(16.0*n^2*R^2)*cdj*cospi*(clj/cdj*tanpi+1.0)*dnpr(j) //
        CT2(i)=CT2(i)+dCT(j)
        CP2(i)=CP2(i)+dCP(j)
        CF2(i)=CF2(i)+dCF(j) //
    end
end
//IMmMp1=M*(I-1)+1
//CT=sum(CT2(IMmMp1:IM))/M
//CP=sum(CP2(IMmMp1:IM))/M
//CF=sum(CF2(IMmMp1:IM))/M //
CT=CT2(IM)
CP=CP2(IM)
CF=CF2(IM)
//
eta=Jv*CT/CP
if (-0.0001<eta)&(eta<0.0001) ieta=%inf else ieta=1/eta end
fm=2/sqrt(pi)*CT^(3/2)/(sqrt(2)*CP)
T=a_rh(H)*n^2*D^4*CT
P=a_rh(H)*n^3*D^5*CP
//
//===== output the file of the aerodynamic characteristics =====
OPf1="I"+string(I)+"M"+string(M)+"N"+string(N)+"BE"+string(round(10*r2d(betv)/10))+ "THE"+...
string(r2d(THE))+ "V"+string(round(V))+ "Vt"+string(round(Vt))+ ". "+ "csv"
ck=clock()
if ck(2)<10 then ckm="0"+string(ck(2)) else ckm=string(ck(2)) end
if ck(3)<10 then ckd="0"+string(ck(3)) else ckd=string(ck(3)) end
if ck(4)<10 then ckh="0"+string(ck(4)) else ckh=string(ck(4)) end
if ck(5)<10 then ckmi="0"+string(ck(5)) else ckmi=string(ck(5)) end
OPf="Results\FVW_" +string(ck(1))+ckm+ckd+ckh+ckmi+"_"+OPf1
Fod=mopen(OPf,"w")
mfprintf(Fod,"%9c,%9c\n","B","R[m]")
mfprintf(Fod,"%5d,%5.3f\n",B,R)
mfprintf(Fod,"%9c\n","b_cpr/R")
for j=1:Jm1
    mfprintf(Fod,"%5.3f",bcpr(j))
end
mfprintf(Fod,"\n%9c\n","b_cpt/R")

```



```

title fontsize 2.0
xlabel("x[m]","fontsize",2.0); ylabel("y[m]","fontsize",2.0); zlabel("z[m]","fontsize",2.0)
subplot(3,2,6); gca().font_size = 2.0
title("THE(tilt angle)=" + string(r2d(THE)) + "deg, " + "Vx=" + string(round(Vx)) + "m/s, " + "Vz=" + string(round(Vz)) + ...
"m/s, " + "V=" + string(round(V)) + "m/s, " + "Vt=" + string(round(Vt)) + "m/s")
title fontsize 2.0
xlabel("x[m]","fontsize",2.0); ylabel("y[m]","fontsize",2.0); zlabel("z[m]","fontsize",2.0)

//
disp("computing time[min]:",toc()/60)
//

```

A.3. サブ・プログラム

(A1) 翼型データの近似式 (FVW_clcd_WS.sce)

```

// Characteristics of the NACA0012, 4412, and RAF6 wing sections.
// 2021.9.9
//
function cll=cl_WS(al, rd, WS)
    alr = d2r(al)
    select WS
    case "NACA0012"
        if al <= -20.0
            cll= 2.0 * sin(alr) * cos(alr)
        elseif al <= -12.0
            cll=-0.0446515*al-1.535818
        elseif al <= 0.0
            cll=-0.000240603*al^3-0.000137123*al^2+0.116201*al
        elseif al <=12.0
            cll=-0.000240603*al^3+0.000137123*al^2+0.116201*al
        elseif al <=20.0
            cll=-0.0446515*al+1.535818
        else
            cll= 2.0 * sin(alr) * cos(alr)
        end
    case "NACA4412"
        if al <= -20.0
            cll = 2.0 * sin(alr) * cos(alr)
        elseif al <= -12.0
            cll=-0.0159015*al-0.960818
        elseif al <= 0.0
            cll=-0.000244301*al^3-0.00322137*al^2+0.0926543*al+0.378965
        elseif al <=13.0
            cll=-0.000186431*al^3+4.967948e-5*al^2+0.107784*al+0.38
        elseif al <=20.0
            cll=-0.105316*al+2.749108
        else
            cll = 2.0 * sin(alr) * cos(alr)
        end
    case "RAF6"
        if al <= -10.0
            cll=2.0*sin(alr)*cos(alr)
        elseif al <= -4.0
            cll=0.062003357*al+0.27801342
        elseif al <=10.0
            cll= -0.00067105688*al^2+0.096294198*al+0.42201664
        elseif al <=30.0
            cll=-0.022698729*al+1.54698729
        else
            cll=2.0*sin(alr)*cos(alr)
        end
    end
endfunction
//
function cdl=cd_WS(al, rd, WS)
    alr = d2r(al)
    select WS
    case "NACA0012"

```

```

if al <= -20.0
  cdl= 2.0 * sin(alr) ^ 2
elseif al <= -8.0
  cdl=0.0008423332*al^2+0.00563866*al+0.0098
elseif al <= 8.0
  cdl=3.90625e-7*al^4+2.98248e-21*al^3+0.0001125*al^2+5.131394e-22*al+0.0098
elseif al <= 20.0
  cdl=0.0008423332*al^2-0.00563866*al+0.0098
else
  cdl= 2.0 *sin(alr) ^ 2
end
// case "NACA0012"
// if al <= -16.0
//   cdl= 2.0 * sin(alr) ^ 2
// elseif al <= -11.0
//   cdl=0.0029533327*al^2+0.052329603*al+0.23317237
// elseif al <= 11.0
//   cdl=6.5051086e-8*al^4+3.246626e-21*al^3+6.6192178e-5*al^2-2.267684e-19*al+0.005932382
// elseif al <= 16.0
//   cdl=0.0029533327*al^2-0.052329603*al+0.2331723
// else
//   cdl= 2.0 * sin(alr) ^ 2
// end
case "NACA4412"
if al <= -20.0
  cdl= 2.0 * sin(alr) ^ 2
elseif al <= -8.0
  cdl=0.000850648*al^2+0.00580518*al+0.0098
elseif al <= 8.0
  cdl=6.012374e-8*al^4+3.464113e-7*al^3+0.000120359*al^2-1.582732e-5*al+0.0098
elseif al <= 20.0
  cdl=0.000850648*al^2-0.00580518*al+0.0098
else
  cdl= 2.0 * sin(alr) ^ 2
end
case "RAF6"
if al <= -10.0
  cdl=2.0*sin(alr)^2
elseif al <= -4.0
  cdl=-0.0055512298*al+0.0047950805
elseif al <= 10.0
  cdl=6.06494e-7*al^5-5.56933e-6*al^4-4.94497e-5*al^3+0.000658908*al^2-0.00040956*al+0.0137695
elseif al <= 30.0
  cdl=0.02345*al-0.2035
else
  cdl=2.0 *sin(alr) ^ 2
end
cdl=cdl+0.28*(t_c(rd)-0.1) // 0.28=(0.05-0.022)/(0.2-0.1)
end
endfunction
//
function tc=t_c(rd)
  tc=-1.50837321414767*rd^3+3.349557320451*rd^2-2.57461505100428*rd+0.788874210540584
endfunction
//

```

(A2) 標準大気 (FVW_atmos.sce)

・参考文献：JIS規格 W 0201-1990 (ISO 2533-1975) (このJIS規格は国内の航空機製造の終了に伴い2016年に廃止された)

```

// Standard Atmosphere
// h[m], a_H(h)[m], a_te(h)[K], a_p(h)[kPa], a_rh(h)[kg/m^3], a_ne(h)[m^2/s], a_so(h)[m/s]
//
global CR Cgm Cg Cra CT0 Cp0 Cr0 Cb1 CHc1 CTc Cb2 CHc2 CS Cbs
CR = 287.05287;
Cgm = 1.4;
Cg = 9.80665;
Cra = 6356766;
CT0 = 288.15;

```

```

Cp0 = 101.325;
Cr0 = 1.225;
Cb1 = -0.0065;
CHc1 = 11000;
CTc = 216.65;
Cb2 = 0.001;
CHc2 = 20000;
CS = 110.4;
Cbs = 0.000001458;
//
function out=a_H(h)
    out = Cra * h / (Cra - h)
endfunction
//
function out=a_te(h)
    z=a_H(h)
    if (0 <= z & z < CHc1) then
        out = CT0 + Cb1 * z
    elseif (CHc1 <= z & z < CHc2) then
        out = CTc
    elseif (CHc2 <= z & z <= 32000) then
        out = CTc + Cb2 * (z - CHc2)
    else
        out = 0
    end
endfunction
//
function out=a_p(h)
    z=a_H(h)
    pc1 = (CTc / CT0) ^ (-Cg / (CR * Cb1)) * Cp0
    pc2 = exp(-Cg / (CR * CTc) * (CHc2 - CHc1)) * pc1
    if (0 <= z & z < CHc1) then
        out = (a_te(h) / CT0) ^ (-Cg / (Cb1 * CR)) * Cp0
    elseif (CHc1 <= z & z < CHc2) then
        out = exp(-Cg / (CR * CTc) * (z - CHc1)) * pc1
    elseif (CHc2 <= z & z <= 32000) then
        out = (a_te(h) / CTc) ^ (-Cg / (Cb2 * CR)) * pc2
    else
        out = 0
    end
endfunction
//
function out=a_rh(h)
    z=a_H(h)
    rc1 = (CTc / CT0) ^ (-Cg / (Cb1 * CR) - 1) * Cr0
    rc2 = exp(-Cg / (CR * CTc) * (CHc2 - CHc1)) * rc1
    if (0 <= z & z < CHc1) then
        out = (a_te(h) / CT0) ^ (-Cg / (Cb1 * CR) - 1) * Cr0
    elseif (CHc1 <= z & z < CHc2) then
        out = exp(-Cg / (CR * CTc) * (z - CHc1)) * rc1
    elseif (CHc2 <= z & z <= 32000) then
        out = (a_te(h) / CTc) ^ (-Cg / (Cb2 * CR) - 1) * rc2
    else
        out = 0
    end
endfunction
//
function out=a_ne(h)
    out = Cbs * a_te(h) ^ 1.5 / (a_te(h) + CS) / a_rh(h)
endfunction
//
function out=a_so(h)
    out=sqrt(Cgm*CR*a_te(h))
endfunction

```

A.4. ブレード形状ファイルの書式

- ・「.csv」ファイルの 2 行目から 11 行目まで、次の項目順番になる。エディターで作成する場合、列の区切りは「,」にな

る。

半径 R [m], ブレード数 B , 接点数 J , 接点位置 r/R , 接点高さ位置 t/R , 代表点位置 r/R , 代表点高さ位置 t/R , 代表点位置の翼弦長 c , 代表点位置のねじり分布ピッチ角 β [deg.], 翼型名

```

• 「FVW_Blade_RAF6_B2e.csv」
2nd:R[m], 3rd:B, 4th:J, 5th:{b_npr/R}, 6th:{b_npt/R}, 7th:{b_cpr/R}, 8th:{b_cpt/R}, 9th:{c/R}, 10th:{BETt[deg.]}, 11th:airfoils
0.457
2
11
0.25 0.325 0.4 0.475 0.55 0.625 0.7 0.775 0.85 0.925 1
0 0 0 0 0 0 0 0 0 0 0
0.2875 0.3625 0.4375 0.5125 0.5875 0.6625 0.7375 0.8125 0.8875 0.9625
0 0 0 0 0 0 0 0 0 0 0
0.1313 0.1501 0.1613 0.166 0.1651 0.159 0.1477 0.1304 0.1063 0.0739
58.9 52.8 47.5 43 39.1 35.8 32.9 30.4 28.3 26.4
RAF6
    
```

```

• 「FVW_Blade_RAF6_B4e.csv」
2nd:R[m], 3rd:B, 4th:J, 5th:{b_npr/R}, 6th:{b_npt/R}, 7th:{b_cpr/R}, 8th:{b_cpt/R}, 9th:{c/R}, 10th:{BETt[deg.]}, 11th:airfoils
0.457
4
11
0.25 0.325 0.4 0.475 0.55 0.625 0.7 0.775 0.85 0.925 1
0 0 0 0 0 0 0 0 0 0 0
0.2875 0.3625 0.4375 0.5125 0.5875 0.6625 0.7375 0.8125 0.8875 0.9625
0 0 0 0 0 0 0 0 0 0 0
0.1313 0.1501 0.1613 0.166 0.1651 0.159 0.1477 0.1304 0.1063 0.0739
58.9 52.8 47.5 43 39.1 35.8 32.9 30.4 28.3 26.4
RAF6
    
```

```

• 「FVW_Blade_0012e.csv」
2nd:R[m], 3rd:B, 4th:J, 5th:{b_npr/R}, 6th:{b_npt/R}, 7th:{b_cpr/R}, 8th:{b_cpt/R}, 9th:{c/R}, 10th:{BET[deg.]}, 11th:airfoils
0.188
2
11
0.25 0.325 0.4 0.475 0.55 0.625 0.7 0.775 0.85 0.925 1
0 0 0 0 0 0 0 0 0 0 0
0.2875 0.3625 0.4375 0.5125 0.5875 0.6625 0.7375 0.8125 0.8875 0.9625
0 0 0 0 0 0 0 0 0 0 0
0.16 0.16 0.16 0.16 0.16 0.16 0.16 0.16 0.16 0.16
0 0 0 0 0 0 0 0 0 0 0
NACA0012
    
```

```

• 「FVW_Blade_0012eD.csv」 (p/D=0.2)
2nd:R[m], 3rd:B, 4th:J, 5th:{b_npr/R}, 6th:{b_npt/R}, 7th:{b_cpr/R}, 8th:{b_cpt/R}, 9th:{c/R}, 10th:{BET[deg.]}, 11th:airfoils
0.188
2
11
0.25 0.325 0.4 0.475 0.55 0.625 0.7 0.775 0.85 0.925 1
0 0 0 0 0 0 0 0 0 0 0
0.2875 0.3625 0.4375 0.5125 0.5875 0.6625 0.7375 0.8125 0.8875 0.9625
0 0 0 0 0 0 0 0 0 0 0
0.16 0.16 0.16 0.16 0.16 0.16 0.16 0.16 0.16 0.16
12.5 10 8.3 7.1 6.2 5.5 4.9 4.5 4.1 3.8
NACA0012
    
```

A.5. 翼型空力特性のプロット確認プログラム (FVW_clcd_WS_plot.sce)

• 「FVW_clcd_WS.sce」で与えた翼型空力特性の近似式関数を、グラフにプロットして特性曲線を確認する。表示グラフは、 $c_l / c_d \sim ALP$ [deg.], および $c_l, c_d \sim ALP$ [deg.]になる。

```

//Plot cl/cd, cl, and cd-ALP[deg.]
//
    
```

```
function rad=d2r(deg)
    rad=deg/180*%pi
endfunction
function deg=r2d(rad)
    deg=rad/%pi*180
endfunction
//
exec('FVW_clcd_WS.sce');
//
WS="NACA0012" // <<====
ALP=45 // <<====
rd=0.75
scf()
for al=-ALP:ALP
    cll=cl_WS(al,rd,WS)
    cdl=cd_WS(al,rd,WS)
    plot2d(al,cll/cdl,-11)
end
xgrid()
ylabel('cl/cd'); ylabel fontsize 2.0; xlabel('ALP [deg.]); xlabel fontsize 2.0
//
scf()
for al=-ALP:ALP
    cll=cl_WS(al,rd,WS)
    plot2d(al,cll,-10)
    cdl=cd_WS(al,rd,WS)
    plot2d(al,cdl,-12)
end
xgrid()
ylabel('cl, cd'); ylabel fontsize 2.0; xlabel('ALP [deg.]); xlabel fontsize 2.0
legend('cl','cd',4)
```

宇宙航空研究開発機構研究開発報告 JAXA-RR-22-001
JAXA Research and Development Report

回転ブレードの自由渦後流・翼素理論による全状態空力特性の概算
Rough Estimation for All-State Aerodynamic Characteristics on Rotating Blades
by Free Vortex Wake and Blade Element Theory

発行 国立研究開発法人 宇宙航空研究開発機構 (JAXA)
〒182-8522 東京都調布市深大寺東町7-44-1
URL: <https://www.jaxa.jp/>

発行日 2022年8月30日
電子出版制作 松枝印刷株式会社

※本書の一部または全部を無断複写・転載・電子媒体等に加工することを禁じます。
Unauthorized copying, replication and storage digital media of the contents of this publication, text and images are strictly prohibited. All Rights Reserved.

